
SfN 2010

We would like to thank everybody who stopped by the CED booth at the Society for
Neuroscience conference in San Diego. The meeting provided an opportunity for the CED
software engineers to meet directly with users to get feedback on the current software and to
discuss ideas to implement in the future. The dynamic clamp features in the forthcoming
Signal version 5 were demonstrated to a number of people with a very positive response and
we had some extremely useful feedback on features we should include in this area. There
was also much interest in the Spike2 Talker interfaces we demonstrated at the meeting from
both users and companies interested in feeding data from their equipment to Spike2.

Talkers

As mentioned in CED Enews #53 Talkers provide a software interface between Spike2 and
other devices without analogue outputs, allowing the system to collect data in addition to that
sampled by a 1401 interface. Talkers have to be written specifically for each type of device
and therefore a degree of programming is required, preferably by the device manufacturer to
whom we can supply comprehensive documentation of the software specifications. If you
know of any devices you feel may be good candidates for a talker interface please let us
know or contact the manufacturers to see if they may be interested in producing the required
software. Typical device types include motion tracking systems, medical devices with a serial
line output and telemetry recording systems.

Future meetings and events

Australian Neuroscience Society
31st Annual Meeting

Auckland, New Zealand
31 Jan - 03 Feb, 2011

Experimental Biology 2011
Washington DC

9-13 April, 2011

Physiology 2011 - Main meeting of the Physiological Society
University of Oxford
11-14 July, 2011

If you are organising or attending a conference at which you feel CED would be interested in
exhibiting please let us know about it by notifying sales@ced.co.uk

Latest versions of Spike2 and Signal

Updates for Spike2 and Signal are available from the CED downloads page, or by clicking on
the links in the table below. Demonstration versions of the latest software are also available.

Spike2 downloads | Released | Signal downloads Released

Spike2 version 7.05 11/10 Signal version 4.08 08/10
Spike2 version 6.15 08/10 Signal version 3.13 05/10

Spike2 demo Signal demo

Back to top


mailto:sales@ced.co.uk
http://www.ced.co.uk/sigdemu.shtml
http://www.ced.co.uk/s2demu.shtml
http://www.ced.co.uk/sigw3u.shtml
http://www.ced.co.uk/s2wu6u.shtml
http://www.ced.co.uk/sigw4u.shtml
http://www.ced.co.uk/s2wu7u.shtml
http://www.ced.co.uk/upu.shtml
http://www.physiology2011.org/
http://experimentalbiology.org/content/default.aspx
http://www.sallyjayconferences.com.au/ans2011/
http://www.ced.co.uk/img/cednws53.pdf

Can | get the 1401 to generate a pulse when it detects a spike during recording?

This can be set up from the on-line Template Setup dialog with the outputs available in
two formats. The first option provides a single pulse from a separate digital bit related
to each of up to 8 spike types but is not useful if more than one channel of spikes is
recorded. The second option provides coded information on the channel number and
spike type using 8 bits of digital outputs. Full details of timing and digital bits used are
available in the Spike2 help.

| perform voltage and current clamp recordings using stepping stimulus pulses. |
overdraw the sweeps but is there any way to make it clear which response relates to
each level of stimulus?

From Signal version 4 the overdraw settings dialog includes a number of options to
fade the colour of each overdrawn sweep from the channel primary colour to another.
The options available are:

1. Fade to background
2. Fade to overdraw colour
3. Fade to secondary colour

With any of these options activated each sweep should be drawn with a slightly
different colour making it easier to visualise channels from the same sweep. Fade to
secondary also allows you to fade to independent colours per channel.

Overdrawn sweeps using 'Fade to secondary colour'

Scripts: Spike2

Q.

A

| am looking for a device ]o generate trains of TTL stimulus pulses but with intervals
randomised around a mean value eg 50ms +/- 20%.

You can achieve this in Spike2 using a combination of a script and output sequence.
The attached script Pulse trains with variability.s2s in combination with
the output sequence Pulse trains with variability.pls enables generation
of this type of pulse train together with some similar and more complex options.

To use this script you must first load the associated output sequence and then run the
script. It displays a toolbar from which you control data sampling and stimulus
generation. Once sampling has started the various stimulus options are enabled.
When you select a stimulus type, a settings dialog will appear in which you specify the
pulse train parameter and when you press OK, the trains will be triggered and a
textmark describing the parameters saved into the data file.

Back to top

Back to top

Back to top

Right-click the script icon and
save.to disk.

If you have any problems
opening the embedded scripts in
this newsletter please let us

know.



'$Pulse trains with variability.s2s|Script to generate trains of pulses with a mean frequency or interval and random variability around the mean

'Requires sequence file Pulse trains with variability.pls loaded into the sampling configuration to run



'The script is a 'work in progress' and is offered without guarantees.

'You must test it to see whether it is suitable for your application.



'CED 24/11/10



var data%;							'Data file handle



var tLength := 45;					'Length of train

var freq := 20;						'Mean frequency of train

var inter := 50;					'Mean interval in ms	

var interval;						'Mean interval in s

var interTrain := 1;                'Interval between trains

var interTrain%;                    'Train interval in sequencer steps

var variability := 20;				'Variability percentage

var inVariability := 0;				'Initial and end variability for ramping variability protocol

var fVariability := 20;

var int[100000];					'Array to fill with interval values

var repeats% := 4;					'Number of repeats

var params$;                        'String to print to textmarks

var code%[4];                       'Code value to textmarks

var com$;

var curSeq$;                        'Name of currently loaded sequencer file





'Set up toolbar

ToolbarSet(1, "Quit");

ToolbarSet(3, "New file", New%);

ToolbarSet(4, "Start", Start%);

ToolbarSet(6, "Textmark", SetTextMark%);

ToolbarSet(8, "Random int train", SetVarIntPulses%);

ToolbarSet(10, "Constant int train", DoLongInt%);

ToolbarSet(12, "Random freq train", SetVarFreqPulses%);

ToolbarSet(14, "Constant Freq train", DoLongFreq%);

ToolbarSet(20, "Stop pulses", StopPulses%);

ToolbarSet(22, "Rand int ramp", SetVarRampInt%);

ToolbarSet(24, "Normal dist int", SetNormInt%);

ToolbarEnable(4,0);

ToolbarEnable(6,0);

ToolbarEnable(8,0);

ToolbarEnable(10,0);

ToolbarEnable(12,0);

ToolbarEnable(14,0);

ToolbarEnable(20,0);

ToolbarEnable(22,0);

ToolbarEnable(24,0);

Toolbar("",1023);





Func New%()                                                 'Function to open new data file for sampling

var ok%;

var errorFlag%;

if SampleHandle(0) > 0 then                                 'Check no file currently open for sampling

    Message("    Data file already open for sampling\nPlease close this down and try again");

    errorFlag% += 1;

    else

    curSeq$ := SampleSequencer$();                          'Check correctly named sequence is loaded

    if Instr(curSeq$,"Pulse trains with variability.pls") = 0 then

        Message("               Incorrect or no sequence file loaded\nPlease load \"Pulse trains with variability.pls\" and try again");

        errorFlag% += 1;

    else

        SampleTextmark(100);

        data% := FileNew(0,3);                              'Open new file

        if data% < 0 then                                   'Check it opened correctly

            Message("                   Failed to start new data file for sampling\nPlease check 1401 is connected and switched on then try again");

        else

            ToolbarEnable(3,0);                             'If successful disable new file button 

            ToolbarEnable(4,1);                             'Enable start button

        endif;

    endif;

endif;

return 1;

end;





Func Start%()

View(data%);

SampleStart();

ToolbarEnable(6,1);

ToolbarEnable(8,1);

ToolbarEnable(10,1);

ToolbarEnable(12,1);

ToolbarEnable(14,1);

ToolbarEnable(20,1);

ToolbarEnable(22,1);

ToolbarEnable(24,1);

ToolbarClear(4);

ToolbarSet(4, "Stop", Stop%);

return 1;

end;





Func Stop%()

View(data%);

SampleStop();

ToolbarClear(4);

ToolbarSet(4, "Start", Start%);

ToolbarEnable(3,1);

ToolbarEnable(4,0);

ToolbarEnable(6,0);

ToolbarEnable(8,0);

ToolbarEnable(10,0);

ToolbarEnable(12,0);

ToolbarEnable(14,0);

ToolbarEnable(20,0);

ToolbarEnable(22,0);

ToolbarEnable(24,0);

return 1;

end;





'Settings for pulse trains with randomised interval varying around a mean

Func SetVarIntPulses%()

var ok%;

DlgCreate("Pulse parameters");

DlgReal(1, "Train duration (s)",1,1000);

DlgReal(2, "Mean Interval (ms)",1,1000);

DlgReal(3, "Interval variability (%)",0,100);

DlgInteger(4, "Number of trains",1,100);

DlgReal(5, "Interval between trains (s)",0.1,60);

ok% := DlgShow(tLength,inter,variability,repeats%,interTrain);

if ok% then

	interval := inter/1000.0;

	GetIntArray(tLength,interval,variability,repeats%,0,interTrain);   'Get array of intervals and pass to sequencer table

	params$ := Print$("Length = %g   Interval = %g   Variability = %g   Trains = %d",tLength,inter,variability,repeats%);   'Print string describing parameters

	ToolbarText(params$);                                              'Show string in toolbar

	SampleKey("a");                                                    'Trigger train in sequence

	code%[0] := 01;                                                    'Set code value for textmark

	SampleText(params$,-1,code%[]);                                    'Generate textmark

endif;

return 1;

end;



'Settings for pulse trains with randomised interval varying around a mean. AMounbt of randomisation varies linearly through time

Func SetVarRampInt%()

var ok%;

DlgCreate("Pulse parameters");

DlgReal(1, "Train duration (s)",1,1000);

DlgReal(2, "Mean Interval (ms)",1,100);

DlgReal(3, "Initial variability (%)",0,100);

DlgReal(4, "Final variability (%)",0,100);

DlgInteger(5, "Number of trains",1,100);

DlgReal(6, "Interval between trains (s)",0.1,60);

ok% := DlgShow(tLength,inter,inVariability,fVariability,repeats%,interTrain);

if ok% then

	interval := inter/1000.0;

	GetVarIntArray(tLength,interval,inVariability,fVariability,repeats%,interTrain);

	params$ := Print$("Length = %g   Freq = %g   Initial var = %g   Final var = %g   Trains = %d",tLength,freq,inVariability,fVariability,repeats%);

	ToolbarText(params$);

	SampleKey("a");

	code%[0] := 06;

	SampleText(params$,-1,code%[]);

endif;

return 1;

end;



'Settings for pulse train with normally distributed randomised interval varying around a mean

Func SetNormInt%()

var ok%;

DlgCreate("Pulse parameters");

DlgReal(1, "Train duration (s)",1,1000);

DlgReal(2, "Mean Interval (ms)",1,1000);

DlgReal(3, "Interval variability (%)",0,100);

DlgInteger(4, "Number of trains",1,100);

DlgReal(5, "Interval between trains (s)",0.1,60);

ok% := DlgShow(tLength,inter,variability,repeats%,interTrain);

if ok% then

	interval := inter/1000.0;

	GetNormIntArray(tLength,interval,variability,repeats%,interTrain);

	params$ := Print$("Length = %g   Interval = %g   Variability = %g   Trains = %d",tLength,inter,variability,repeats%);

	ToolbarText(params$);

	SampleKey("a");

	code%[0] := 01;

	SampleText(params$,-1,code%[]);

endif;

return 1;

end;



'Settings for pulse train with randomised frequency varying around a mean

Func SetVarFreqPulses%()

var ok%;

DlgCreate("Pulse parameters");

DlgReal(1, "Train duration (s)",1,1000);

DlgReal(2, "Mean frequency",1,100);

DlgReal(3, "Frequency variability (%)",0,100);

DlgInteger(4, "Number of trains",1,100);

DlgReal(5, "Interval between trains (s)",0.1,60);

ok% := DlgShow(tLength,freq,variability,repeats%,interTrain);

if ok% then

	GetIntArray(tLength,freq,variability,repeats%,1,interTrain);

	params$ := Print$("Length = %g   Freq = %g   Variability = %g   Trains = %d",tLength,freq,variability,repeats%);

	ToolbarText(params$);

	SampleKey("a");

	code%[0] := 02;

	SampleText(params$,-1,code%[]);

endif;

return 1;

end;







Func StopPulses%()

View(data%);

SampleKey("h");                                         'Set key to stop sequence

code%[0] := 00;

SampleText("Stopped pulses",MaxTime(),code%[]);

return 1;

end;



'Function to generate array of intervals to pass to sequencer table

Func GetIntArray(len,it,vari,rep%,type%,trainInt)

var num% := 0;

var tInt;

var minInt;

var intScale;

var varInt;

var maxFreq;

var fScale;

var varFreq;

if type% = 0 then                                       'If working in intervals

	intScale := (vari/100.0)*it*2;                      'Calculate scale and offset values for us in random number generator

	minInt := it - (it * ((vari/100.0)));

	repeat                                              'Start loop

		varInt := Rand(intScale,minInt);                'Get random interval value in seconds

		int[num%] := varInt;                            'Add to array

		tInt += int[num%];                              'Calculate total time

		num% += 1;                                      'Add 1 to count of intervals

	until tInt >= len;                                  'Repeat until total train time reached

    interTrain% := Round(trainInt*100000)-3;            'Convert inter train interval to sequencer steps

	ArrMul(int[:num%],100000);

	Round(int[:num%]);

	var int1%[num%];									'Create integer array to pass to sequencer

	ArrConst(int1%[],int[:num%]);                       

	ArrSub(int1%[],5);                                  'Subtract 5 steps to take into account sequencer loop

	SampleSeqTable(int1%[:num%]);                       'Send to sequencer table

	SampleSeqVar(2,num%-1);                             'Set number of steps in sequence

	SampleSeqVar(3,rep%);                               'Set number of trains in sequence

    SampleSeqVar(8,interTrain%);                        'Set inter train interval in sequence

else

	fScale := it * (vari/100.0)*2;                      'If working in frequency calculate required scale and offset 

	maxFreq := it + (it * ((vari/100.0)));

	repeat                                              'Start loop

		varFreq := maxFreq - Rand(fScale,0);            'Get random frequency value

		int[num%] := 1.0/varFreq;                       'Convert to interval

		tInt += int[num%];                              'Add to total time

		num% += 1;                                      'Add to event count

	until tInt >= len;                                  'Loop until train time reached

    interTrain% := Round(trainInt*100000)-3;            'Convert inter train interval to sequencer steps

	ArrMul(int[:num%],100000);

	Round(int[:num%]);

	var int2%[num%];									'Create integer array to pass to sequencer

	ArrConst(int2%[],int[:num%]);

	ArrSub(int2%[],5);

	SampleSeqTable(int2%[:num%]);                       'Send to sequencer table

	SampleSeqVar(2,num%-1);                             'Set number of steps in sequence

	SampleSeqVar(3,rep%);                               'Set number of trains in sequence

    SampleSeqVar(8,interTrain%);                        'Set inter train interval in sequence

endif;

return 1;

end; 



'Function to generate randomised interval train with linear increase in variability

Func GetVarIntArray(len,it,sVari,eVari,rep%,trainInt)

var num% := 0;

var tInt := 0;

var minInt;

var intScale;

var varInt;

var maxFreq;

var fScale;

var varFreq;

var cVari;

repeat                                                  'Start loop

	cVari := sVari + ((tInt/len) * (eVari - sVari));    'Get current level of randomisation based on time relative to start of train

	intScale := (cVari/100.0)*it*2;                     'calculate scale and offset at this time for random number generator

	minInt := it - (it * ((cVari/100.0)));

	varInt := Rand(intScale,minInt);                    'Get randomised interval

	int[num%] := varInt;                                'Add to array

	tInt += int[num%];                                  'Get total time

	num% += 1;                                          'Add to count of intervals

until tInt >= len;                                      'Loop until train time reached

'ArrConst(int[num%-1:4],0.25);						'Set 3 values at 0.25s interval to separate trains by 1 second 

interTrain% := Round(trainInt*100000)-3;            'Convert inter train interval to sequencer steps

ArrMul(int[:num%],100000);

Round(int[:num%]);

var int1%[num%];									'Create integer array to pass to sequencer

ArrConst(int1%[],int[:num%]);

ArrSub(int1%[],5);

SampleSeqTable(int1%[:num%]);                       'Send to sequencer table

SampleSeqVar(2,num%-1);                             'Set number of steps in sequence

SampleSeqVar(3,rep%);                               'Set number of trains in sequence

SampleSeqVar(8,interTrain%);                        'Set inter train interval in sequence

return 1;

end;



'Function to generate array of random intervals around mean with normal distribution

Func GetNormIntArray(len,it,vari,rep%,trainInt)

var num% := 0;

var tInt;											'Total time

var minInt;											'Minimum interval

var intScale;										'Scale factor for interval variation calculation

var varInt;											'randomly generated interval value

	intScale := (vari/100.0)*it/3;                  'Calculate scale and offset for random number generator

	minInt := it;

	repeat                                          'Start loop

		varInt := RandNorm(intScale,minInt);        'Get random value for interval in seconds

		int[num%] := varInt;                        'Add to array

		tInt += int[num%];                          'Get total time

		num% += 1;                                  'Add to interval count

    until tInt >= len;

    interTrain% := Round(trainInt*100000)-3;            'Convert inter train interval to sequencer steps

	ArrMul(int[:num%],100000);

	Round(int[:num%]);

	var int1%[num%];									'Create integer array to pass to sequencer

	ArrConst(int1%[],int[:num%]);

	ArrSub(int1%[],5);

	SampleSeqTable(int1%[:num%]);                       'Send to sequencer table

	SampleSeqVar(2,num%-1);                             'Set number of steps in sequence

SampleSeqVar(3,rep%);                                   'Set number of trains in sequence

SampleSeqVar(8,interTrain%);                            'Set inter train interval in sequence

return 1;

end;





Func SetTextMark%()

var tTime;

View(data%);

tTime := MaxTime();

DlgCreate("Add textmark");

DlgString(1,"Comment",100);

DlgInteger(2,"Code",0,255);

DlgShow(com$,code%[0]);

SampleText(com$,tTime,code%[]);

return 1;

end;





Func DoLongFreq%();

var freq := 30;

var len := 10;

var ok%;

DlgCreate("Train paramaters");

DlgReal(1,"Frequency",1,100);

DlgReal(2,"Duration (s)",1,1000);

DlgAllow(1023);

ok% := DlgShow(freq,len);

if ok% then

	var num%;

	var delSteps%;

	num% := Round(freq * len) + 1;

	delSteps% := ((1/freq) * 100000) - 4;

	SampleSeqVar(4,num%);

	SampleSeqVar(5,delSteps%);

	params$ := Print$("Length = %g   Frequency = %g",len,freq);

	ToolbarText(params$);

	SampleKey("a");

	code%[0] := 04;

	SampleText(params$,-1,code%[]);

	SampleKey("l");

endif;

return 1;

end;





Func DoLongInt%();

var int := 30;

var len := 10;

var ok%;

DlgCreate("Train paramaters");

DlgReal(1,"Interval (ms)",1,100);

DlgReal(2,"Duration (s)",1,1000);

DlgAllow(1023);

ok% := DlgShow(int,len);

if ok% then

	var num%;

	var delSteps%;

	num% := len/(int/1000.0) + 1;

	delSteps% := (int * 100) - 4;

	SampleSeqVar(4,num%);

	SampleSeqVar(5,delSteps%);

	params$ := Print$("Length = %g   Interval = %g",len,int);

	ToolbarText(params$);

	SampleKey("a");

	code%[0] := 03;

	SampleText(params$,-1,code%[]);

	SampleKey("l");

endif;

return 1;

end;

CED
Pulse trains with variability.s2s




CED
Pulse trains with variability.pls

CED
HINT: You can generate random intervals between pulses using the graphical sequence editor. Amongst the 'Wait for time, condition or variable' options you can add to the control line are random and poisson delay.  


The types of pulse train included are:

Random int train - Number of pulse trains in which you can specify mean
interval and percentage of random variation around the mean as well as number
of trains and interval between trains.

Constant int train - Single pulse train with constant interval.

Random freq train - Similar to random int train although based on
frequency rather than interval.

Constant freqg train - Single pulse train with constant frequency.

Random int ramp - Number of pulse trains with mean interval but the amount
of random variation ramps linearly through time.

Normal dist int - Number of pulse trains where you specify the mean

interval and an amount of variation but this is normally distributed rather than
purely random.

i N

Pulse parameters

Train duration [z] |457
bean Interval [ms) |5EI7
Initial variability (%] b
Final variabilty (%] 20
MHumber of traing |4=
Interval between traing [=] #

|Eanu:e|| | ok |

e E - )

Example pulse train parameter dialog

Scripts: Signal

Q:

| record single channel patch clamp data and use the SCAN method in Signal to
generate idealised traces. | export the idealised trace as text which gives me most of
the required information including a flag value, however | would like to know what
these flags represent (eg open, closed, first latency etc.).

The attached script Open closed event time printout.sgs prints a text file in
a similar format to the standard text export but following the flag value prints the actual
meanings of the flag value for each event. To use it, simply open a Signal data file
containing idealised trace data then run the script. It currently prints all of the data for
each frame but could easily be modified to allow you to select a frame subset and time
range if required.

Frame Time Dur Amp Flags

1 0.000000  0.017500 -0.137457 80 Closed time. First latency
1 0.017500  0.004100 13.446917 128 Open time.

1 0.021600  0.001300 -0.136150 64 Closed time.

1 0.022900  0.000900 11.865234 128 Open time.

Typical open/closed time event data

Back to top

Right-click the script icon and
save to disk.

If you have any problems
opening the embedded scripts in
this newsletter please let us
know.



'$Open closed event time printout.sgs|Signal script to print idealised trace open/closed time information to a text file



'Requires a Signal data file containing at least one idealised trace to be open and set to be the

' current view before running



'The script is a 'work in progress' and is offered without guarantees.

'You must test it to see whether it is suitable for your application.



'CED 25/11/10





Var data%;  'Data file handle

var idChans%[100];                      'Array to hold idealised trace channels

var chan%;                              'Selected idealised trace channel to export

var ok%;



Var txt%; 'Text file handle





Data%:=FrontView();                     'Get handle of current view

If Viewkind(data%)<>0 then              'Continue if it is a data file

    Message("View is not a data view. Halting");

    halt

Endif



ChanList(idChans%[],4);                 'List all idealised traces

docase

case idChans%[0] = 0 then               'Check if any idealised traces in file

    Message("No idealised trace channels present. Halting");

    Halt;

case idChans%[0] = 1 then               'If just single idealised trace, use it

    chan% := idChans%[1];

case idChans%[0] > 1 then               'If more than one, generate dialog for user to select channel

    DlgCreate("Idealised trace export");

    DlgChan(1, "Select channel",4);

    ok% := DlgShow(chan%);

    if ok% = 0 then                     'Proceed only if user presses OK

        Message("Pressed cancel, quitting script");

        Halt;

    endif;

endcase





Setup%();

PrintEvents%();





Func setup%();

Txt%:=Filenew(1);

If txt%<0 then

    Message("%s HALTING", Error$(txt%));

    halt

Endif

WindowTitle$(View(data%).Filename$(3)+filename$(4)+"txt");



Return 1

end



Func Printevents%();

Var Time, LastT;

Var Dur, Amp, flags%, Flags$;

Var num%;

Var err%;

Var i%;

Var FlagsArr$[16];

Arrconst(FlagsArr$[1:Len(flagsArr$[])-2], "\t");



View(data%);

View(txt%).Print("\nFrame\tTime\tDur\tAmp\tFlags");





For i%:=1 to FrameCount() do

    Frame(i%);

    Time:=Mintime();

    lastt:=Time;

    'Printlog("\nFrame %d",i%);

    repeat

        Err%:=OpClEventGet(chan%, 2, time, Dur, amp, flags%);

        

        If Time < lastT then

            err%:=0;

        Endif        

        

        if err%>0 then        

            Num%:=Flags%;

            

            

            Repeat

                docase

                Case Flags%>= 32768 then

                    Flags$:= Flags$+"\tLevel 2.";

                    flags%:=Flags%-32768;

                Case Flags%>= 16384 then

                    Flags$:= Flags$+"\tLevel 3.";

                    Flags%:= Flags%-16384;            

                Case Flags%>= 8192 then

                    Flags$:= Flags$+"\tLevel 4.";

                    Flags%:= Flags%-8192;

                Case Flags%>= 4096 then

                    Flags$:= Flags$+"\tLevel 5.";            

                    Flags%:= Flags%-4096;

                Case Flags%>= 2048 then

                    Flags$:= Flags$+"\tLevel 6.";

                    Flags%:= Flags%-2048;

                Case Flags%>= 1024 then

                    Flags$:= Flags$+"\tReserved"; 

                    Flags%:= Flags%-1024;

                Case Flags%>= 512 then

                    Flags$:= Flags$+"\tReserved"; 

                    Flags%:= Flags%-512;

                Case Flags%>= 256 then

                    Flags$:= Flags$+"\tReserved"; 

                    Flags%:= Flags%-256;

                Case Flags%>= 128 then

                    Flags$:= Flags$+"\tOpen time.";

                    Flags%:= Flags%-128;

                Case Flags%>= 64 then

                    Flags$:= Flags$+"\tClosed time.";            

                    Flags%:= Flags%-64;

                Case Flags%>= 32 then

                    Flags$:= Flags$+"\tTruncated: The last event in an idealised trace.";

                    Flags%:= Flags%-32;

                Case Flags%>= 16 then

                    'Flags$:= Flags$+"\tFirst latency: The period from the start of the idealised trace to the first transition.";            

                    Flags$:= Flags$+"\tFirst latency";            

                    Flags%:= Flags%-16;

                Case Flags%>= 8 then

                    Flags$:= Flags$+"\tReserved";

                    Flags%:= Flags%-8;

                Case Flags%>= 4 then

                    Flags$:= Flags$+"\tAssumed amplitude: Events whose amplitude has not been calculated from the raw data." ;

                    Flags%:= Flags%-4;

                Case Flags%>= 2 then

                    Flags$:= Flags$+"\tBad data.";         

                    Flags%:= Flags%-2;

                Case Flags%>= 1 then

                    'Flags$:= Flags$+"\tLevel 1. Closed times and the first open level will have this set.";

                    Flags$:= Flags$+"\tLevel 1.";

                    

                    Flags%:= Flags%-1;

                endcase  

                   

            Until Flags%=0;

            

            View(txt%).Print("\n%d\t%f\t%f\t%f\t%D\t%S", i%,Time, Dur, Amp, Num%, Flags$);

            

            Flags$:="";

            

            Time:=Time+Dur;

            Lastt:=Time;

        endif 

    Until Err%=0;

next 

    View(Txt%);

    Window(50,0,100,100);

    WindowVisible(1);

    Frontview();

Message("Done");



Return 1

end

CED
Open closed event time printout.sgs


Did you know...?

You can control the display of x and y grid lines independently through the Show Hide e

Channel dialog or with the Grid () script command.

Time view with grid displayed only on the y axis

Recent questions Back to top

Q. | am using the measure to channel function to detect times of peaks in a blood
pressure channel on-line. The peak detection function for cursor 0 stepping requires
an amplitude value but as the pulse pressure (peak to peak amplitude) varies | would
like to be able to adjust this during sampling. Is there a way to do this without using a
script.

A. Horizontal cursor positions can be used in this case. If you place two horizontal
cursors on the pressure channel trace you can set the amplitude in the peak detection
option in Cursor 0 stepping to use the level difference between the cursors. If for
example you use horizontal cursor 1 above 2 you can use the notation H1-H2. During
the recording you can then adjust the cursors to be closer or further apart to
compensate for changes in pulse pressure.

Detected arterial pressure peaks using an amplitude set by H1-H2

User group
You can now find our eNewsletters archived on the CED website.
Try the CED Forums bulletin board for software and hardware support

If you have any comments about the newsletter format and content, or wish to un-subscribe
from the mailing list for this newsletter, please notify sales@ced.co.uk.


mailto:sales@ced.co.uk?subject=eNews%20Un-sunscribe
http://www.ced.co.uk/phpBB3/index.php
http://www.ced.co.uk/nwenwu.htm
CED
HINT: You can set the large tick spacing and number of subdivisions displayed on the axes and grid from the x and y axis range dialogs




